Fourier Spectrum, Power Spectrum and Autocorrelation—FPAC

The program FPAC (**F**ourier Spectrum, **P**ower Spectrum and **A**uto**c**orrelation) is a subroutine subprogram that simultaneously computes any two or all of the Fourier spectrum, power spectrum and autocorrelation functions for a given set of equally spaced data. The argument *IND* specifies which one to compute.

FPAC（**F**ourier Spectrum, **P**ower Spectrum and **A**uto**c**orrelation）

【Purpose】

To compute any two or all of the Fourier spectrum, power spectrum, and autocorrelation functions for the given equally spaced data.

【Usage】

（1） How to connect

```
CALL FPAC (N, X, ND1, DT, IND, F, G, R, ND2, NFOLD, DF)
```

Argument	Type	Parameter in calling program	Return Parameter
N	I	Total number of real data X	Unchanged
X	R 1-D array (ND1)	Equal interval real data	Unchanged
ND1	I	Dimension size of X in calling program (ND1 .LE. 8192)	Unchanged
DT	R	Time interval (unit : sec)	Unchanged
IND	I	Index for calculation 100 : Fourier Spectrum 010 : Power Spectrum 001 : Autocorrelation Function	Unchanged
F	R 1-D array (ND2)	No need to input here	Fourier Spectrum
G	R 1-D array (ND2)	No need to input here	Power Spectrum
R	R 1-D array (ND2)	No need to input here	Autocorrelation Function

ND2	I	Dimension size of F, G and R in calling program	Unchanged
NFOLD	I	No need to input here	Total number of Fourier & Power Spectra and Autocorrelation Function
DF	R	No need to input here	Frequency interval of Fourier & Power Spectra (Unit : Hz)

(2) Necessary subroutines and function subprograms

FAST

(3) Remarks

i) The argument *IND* can be added together. For example, when *IND*=101(100+001), the Fourier spectrum and the autocorrelation function are computed. The results for all combinations of *IND* are shown in the table below. The areas marked with an asterisk (*) are empty when you return from the program.

IND	*F*	*G*	*R*
111	○	○	○
110	○	○	*
101	○	*	○
100	○	*	*
011	○	○	○
010	○	○	*
001	*	*	○

ii) *ND*2 must be greater than or equal to *NT*/2+1, where *NT* is the smallest power of 2 greater than *N*, or *N* if *N* is a power of 2.

iii) The values of the autocorrelation function are normalized by the mean square of the data, and are given at intervals of time shift equal to the time interval *DT* of the data.

【Program List】

```
C  * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * *      FPAC   1
C     SUBROUTINE FOR FOURIER, POWER SPECTRA AND AUTOCORRELATION           FPAC   2
C  * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * *      FPAC   3
C                                                                          FPAC   4
C                                                CODED BY Y.OHSAKI         FPAC   5
C                                                                          FPAC   6
C     PURPOSE                                                              FPAC   7
C       TO COMPUTE FOURIER SPECTRUM, POWER SPECTRUM AND/OR AUTOCORRE-      FPAC   8
C       LATION OF A SERIES OF EQUI-SPACED DATA                             FPAC   9
C                                                                          FPAC  10
C     USAGE                                                                FPAC  11
C       CALL FPAC(N,X,ND1,DT,IND,F,G,R,ND2,NFOLD,DF)                       FPAC  12
C                                                                          FPAC  13
C     DESCRIPTION OF AUGUMENTS                                             FPAC  14
C       N      - TOTAL NUMBER OF DATA                                      FPAC  15
C       X(ND1) - EQUI-SPACED DATA                                          FPAC  16
```

```
C       ND1    - DIMENSION OF X IN CALLING PROGRAM  ND1.LE.8192          FPAC  17
C       DT     - TIME INCREMENT IN DATA IN SEC                           FPAC  18
C       IND    - 100 FOR FOURIER SPECTRUM                                FPAC  19
C                010 FOR POWER SPECTRUM                                  FPAC  20
C                001 FOR AUTOCORRELATION                                 FPAC  21
C       F(ND2) - FOURIER SPECTRUM                                        FPAC  22
C       G(ND2) - POWER SPECTRUM                                          FPAC  23
C       R(ND2) - AUTOCORRELATION                                         FPAC  24
C       ND2    - DIMENSION OF F,G,R IN CALLING PROGRAM                   FPAC  25
C       NFOLD  - TOTAL NUMBER OF DATA IN SPECTRUM AND AUTOCORRELATION    FPAC  26
C       DF     - FREQUENCY INCREMENT IN FOURIER AND POWER SPECTRA IN HZ   FPAC  27
C                                                                        FPAC  28
C     REMARKS                                                            FPAC  29
C      (1) PARAMETER IND IS ADDIBLE. IF, FOR INSTANCE, IND=101(100+001),FPAC  30
C          FOURIER SPECTRUM AND AUTOCORRELATION ARE COMPUTED             FPAC  31
C      (2) ND2.GE.NT/2+1, WHERE NT IS POWER OF 2 EQUAL TO N OR MINIMUM   FPAC  32
C          LARGER THAN N                                                 FPAC  33
C      (3) AUTOCORRELATION IS NORMALIZED IN TERMS OF THE MEAN OF DATA    FPAC  34
C          SQUARED                                                       FPAC  35
C      (4) AUTOCORRELATION IS SPACED AT THE SAME INTERVAL AS THE GIVEN   FPAC  36
C          SERIES OF DATA                                                FPAC  37
C                                                                        FPAC  38
C     SUBROUTINES AND FUNCTION SUBPROGRAMS REQUIRED                      FPAC  39
C       FAST                                                             FPAC  40
C                                                                        FPAC  41
      SUBROUTINE FPAC(N,X,ND1,DT,IND,F,G,R,ND2,NFOLD,DF)                 FPAC  42
C                                                                        FPAC  43
      COMPLEX   C(8192)                                                  FPAC  44
      DIMENSION X(ND1),F(ND2),G(ND2),R(ND2)                              FPAC  45
C                                                                        FPAC  46
C     INITIALIZATION                                                     FPAC  47
C                                                                        FPAC  48
      DO 110 M=1,N                                                       FPAC  49
      C(M)=CMPLX(X(M),0.)                                                FPAC  50
  110 CONTINUE                                                           FPAC  51
      NT=2                                                               FPAC  52
  120 IF(NT.GE.N) GO TO 130                                              FPAC  53
      NT=NT*2                                                            FPAC  54
      GO TO 120                                                          FPAC  55
  130 IF(NT.EQ.N) GO TO 150                                              FPAC  56
      DO 140 M=N+1,NT                                                    FPAC  57
      C(M)=(0.,0.)                                                       FPAC  58
  140 CONTINUE                                                           FPAC  59
  150 NFOLD=NT/2+1                                                       FPAC  60
      T=REAL(NT)*DT                                                      FPAC  61
      DF=1./T                                                            FPAC  62
C                                                                        FPAC  63
C     FOURIER TRANSFORM                                                  FPAC  64
C                                                                        FPAC  65
      CALL FAST(NT,C,8192,-1)                                            FPAC  66
C                                                                        FPAC  67
C     FOURIER SPECTRUM                                                   FPAC  68
C                                                                        FPAC  69
```

```
      IF(IND.EQ.001) GO TO 180                                          FPAC  70
      DO 160 K=1,NFOLD                                                  FPAC  71
      F(K)=CABS(C(K))*DT                                                FPAC  72
  160 CONTINUE                                                          FPAC  73
      IF(IND.EQ.100) RETURN                                             FPAC  74
C                                                                       FPAC  75
C     POWER SPECTRUM                                                    FPAC  76
C                                                                       FPAC  77
      IF(IND.EQ.101) GO TO 180                                          FPAC  78
      G(1)=F(1)**2/T                                                    FPAC  79
      DO 170 K=2,NFOLD-1                                                FPAC  80
      G(K)=2.*F(K)**2/T                                                 FPAC  81
  170 CONTINUE                                                          FPAC  82
      G(NFOLD)=F(NFOLD)**2/T                                            FPAC  83
      IF(MOD(IND,10).EQ.0) RETURN                                       FPAC  84
C                                                                       FPAC  85
C     AUTOCORRELATION                                                   FPAC  86
C                                                                       FPAC  87
  180 DO 190 K=1,NT                                                     FPAC  88
      C(K)=C(K)*CONJG(C(K))                                             FPAC  89
  190 CONTINUE                                                          FPAC  90
      CALL FAST(NT,C,8192,+1)                                           FPAC  91
      R0=REAL(C(1))                                                     FPAC  92
      DO 200 J=1,NFOLD                                                  FPAC  93
      R(J)=REAL(C(J))/R0                                                FPAC  94
  200 CONTINUE                                                          FPAC  95
      RETURN                                                            FPAC  96
      END                                                               FPAC  97
```

【Example】

Calculate the Fourier spectrum, power spectrum, and autocorrelation function of El Centro seismic wave (EQ.01).

```
C
      DIMENSION DATA(800),F(513),G(513),r(513)
C
      READ(5,501) DT,NN,(DATA(M),M=1,NN)
      CALL FPAC(NN,DATA,800,DT,111,F,G,R,513,NFOLD,DF)
      STOP
  501 FORMAT(T51,F10.0,I10/(8F10.0))
      END
```

Output：

The resultant Fourier spectrum, power spectrum, and autocorrelation function are stored in the arrays *F*, *G*, and *R*, respectively. They are shown in the next page.


Fourier Spectrum (cm/sec)
200
150
100
50
0
0 1 2 3 4 5 6 7 8 9 10
Frequency (Hz)


Power Spectrum (cm²/sec³)
6,000
5,000
4,000
3,000
2,000
1,000
0
0 1 2 3 4 5 6 7 8 9 10
Frequency (Hz)


Autocorrelation Coefficient
1.0
0.8
0.6
0.4
0.2
0.0
-0.2
-0.4
0 1 2 3 4
Time Shift (sec)